Coordinate Transformation from the Moon Mean Equator and IAU Node of J2000 System to the Lunar PA Coordinate System

According to "Report of the IAU/IAG Working Group on Cartographic Coordinates and Rotational Elements: 2006", the transformation matrix from the Earth mean equator and equinox of J2000 (EME2000) coordinate system to the moon-centered, body-fixed lunar principal axis (PA) system is given by the following (3-1-3) rotation sequence;

$$[M]^{PA}_{EME2000} = R_z(\psi)\,R_x(\theta)\,R_z(\varphi)$$

In this equation, φ is the angle along the ICRF (International Celestial Reference Frame) equator, from the ICRF x-axis to the ascending node of the lunar equator, θ is the inclination of the lunar equator to the ICRF equator, and ψ is the angle along the lunar equator from the node to the lunar prime meridian. These three Euler angles represent the numerically integrated physical librations of the Moon.

The principal axis frame is aligned with the three maximum moments of inertia of the Moon. The LP150Q lunar gravity model was developed using this frame.

The relationship between these angles and the classical IAU (International Astronomical Union) orientation angles is

$$\alpha = \varphi - 90^\circ$$

$$\delta = 90^\circ - \theta$$

$$W = \psi$$

For long-term orbit prediction of low lunar orbits (LLO), we require the transformation between the inertial Moon mean equator and IAU node of J2000 system and the body-fixed PA coordinate system.

The transformation from this inertial system which we will call moon_j2000 to the PA system is given by the following matrix multiplication;

$$[N]^{PA}_{moon\_j2000} = [M]^{PA}_{EME2000}\,[P]^{EME2000}_{moon\_j2000}$$

The numerical components of the constant moon_j2000-to-EME200 transformation matrix are as follows;

```
 0.998496505205088      4.993572939853833E-2   -2.260867140418499E-2
-5.481540926807404E-2   0.909610125238044      -0.411830900942612
 0.000000000000000      0.412451018902688       0.910979778593430
```

For accurate orbital analysis, the lunar libration angles can be obtained from a lunar ephemeris. The JPL Developmental Ephemeris DE403 is considered the best currently available lunar ephemeris. Chebyshev polynomial representations of the Euler angles and their rates in the PA system are included in the DE403 ephemeris file which is available from http://ssd.jpl.nasa.gov.

Fortran subroutine

The following is the source code listing for a Fortran subroutine that evaluates the DE403 ephemeris file and computes this coordinate transformation.

```
      subroutine moon_pa(xjdate, tmatrix)

c     moon_j2000-to-moon_pa transformation
c     matrix using DE403.bin binary ephemeris

c     input

c      xjdate = TDB julian date

c     output

c      tmatrix = moon_j2000-to-moon_pa transformation matrix

c     Orbital Mechanics with Fortran

c     **********************************

      implicit double precision (a-h, o-z)

      dimension sv(6), tmatrix1(3, 3), tmatrix2(3, 3)

      dimension tmatrix(3, 3)

c     moon_j2000-to-j2000 transformation matrix

      data tmatrix2
     &           /  0.998496505205088d0,
     &             -5.481540926807404d-2,
     &              0.0d0,
     &              4.993572939853833d-2,
     &              0.909610125238044d0,
     &              0.412451018902688d0,
     &             -2.260867140418499d-2,
     &             -0.411830900942612d0,
     &              0.910979778593430d0 /

c     compute lunar libration angles (radians)

      icent = 0

      itarg = 15

      call jpleph (xjdate, itarg, icent, sv)
```

```
      phi = sv(1)

      theta = sv(2)

      psi = sv(3)

c     compute lunar libration matrix

      call matran (phi, 3, theta, 1, psi, 3, 0.0d0, 0, tmatrix1)

c     create moon_j2000-to-moon_pa transformation matrix

      call matxmat (tmatrix1, tmatrix2, tmatrix, 3, 3, 3)

      end
```

In this subroutine, `jpleph` is a subroutine that evaluates the binary ephemeris file, `matran` is a subroutine that performs a sequence of ordered rotations, and `matxmat` is a Fortran subroutine that performs multiplies two matrices. The source code for the `jpleph` subroutine is provided by JPL as are routines for creating a binary DE403 ephemeris file from the ASCII source data. It also includes routines that open and read the binary file.